
 

 

Contents 
News – Updates & meetings  
Spike2 – Fitting a Gabor function 
Signal –Introduction to multiple states 
Spike2 Script – Artefact removal 
Signal Script – Simple channel arithmetic 
Did you know…? – Filter banks 
Recent questions – Detect half-height of spikes 

News  

Spike2 and Signal Training Days 
This years U.K. Spike2 and Signal training courses are fast approaching and will take 
place on Thursday 30th April and Friday 1st of May in Cambridge. Places are filling up 
quickly so If you or any of your colleagues would like to attend, please follow the links to 
view the training day program and complete the registration form. Please give a short 
description of the type of research that you do and analysis that you are interested in. 
This will allow us to tailor sessions for specific requirements. 

Training Days in China 
We will also be running our first Chinese training days in Shanghai during the week 
beginning the 6th of April. These training days will be run in association with Qichi 
Instruments, distributors of CED equipment in China. For further information please 
contact Mr. Ryan Huang at ryan@qichi-instruments.com 

Meetings and events 
Biophysical Society Annual Meeting 
Boston Convention and Exhibition Center  
Boston, MA, USA 
March 1st – 3rd 2009 

16th Annual Meeting of the Cognitive Neuroscience Society 
Hyatt Regency San Francisco  
San Francisco, CA, USA 
March 21st – 24th 2009 

German Neuroscience Society  
Goettingen Exhibition Centre   
Goettingen, Germany 
March 26th – 29th 2009 

Experimental Biology 2009 
Ernest N. Morial Convention Center  
New Orleans, LA, USA 
April 18th – 22nd 2009 

British Neuroscience Association 20th National Meeting 
Adelphi Hotel  
Liverpool, UK 
April 19th – 22nd 2009 

 

 

International Workshop and Conference on Human Reflexes: 
Wiring and Firing of Motorneurons 
Ege University  
Izmir, Turkey 
May 11th – 15th 2009 

The 6th Congress of the International Society for Autonomic 
Neuroscience 
ISAN09 is a joint meeting to be held with the European 
Federation of Autonomic Societies (EFAS) in Sydney, from 
September 1st-4th, 2009. Approx. 400 international and national 
delegates are expected, having clinical and/or basic scientific 
research interests in autonomic neuroscience. 

Plenary speakers: Darwin Berg (UCSD), Hugo Critchley (University of Sussex), Max Hilz 
(University of Erlangen-Nuremberg) and Elspeth McLachlan (POWRI) 

Further details of the meeting can be found at http://www.iceaustralia.com/isan2009/ 

Proudly supported by Cambridge Electronic Design and the Foundation for High Blood 
Pressure Research, Australia. 

Latest versions of Spike2 and Signal 
Updates for Spike2 and Signal are available from the CED downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software are 
also available.   

 Spike2 downloads Signal downloads 
Spike2 version 6.10 Signal version 4.04 

Spike2 version 5.20 Signal version 3.11 

Spike2 demo Signal demo 

 

Spike2   
Q. I am using Spike2 to record multi-unit spike activity and generate auto-

correlograms with the intention of investigating oscillating neuronal activity. This 
has been done in the past by fitting a Gabor function to auto-correlogram data. Is 
it possible to fit a Gabor function in Spike2? 

A. The attached script, GaborFit.s2s is an example of curve-fitting by non-linear 
regression. The Gabor function used is a sine wave damped by a Gaussian 
function. The script allows the user to generate suitable test data and select 
result view files for fitting. 

 

  

 

Q. I’ve been told by a colleague that Signal has a built-in system of controlling 
multiple stimulus protocols during an experiment. For instance, applying a set of 
five different paired stimulus pulses in a random order. Can you give me any 
further information on using Signal in this way?   

A. Multiple states sampling is one of the most powerful features in Signal. It is 
widely used in TMS, evoked response and patch and voltage clamp studies 
where it is common to setup a number of different sets of output pulses for use 
during an experiment and switch between these outputs during sampling.  

 As an example, you could set up one state that generates a single stimulation 
pulse on a DAC output, another state that has two stimulation pulses separated 
by 20 milliseconds and a third state containing a pulse train consisting of 10 
pulses at 20Hz. Signal can then switch between these outputs randomly or in a 
preset order during sampling. Every sampled frame stores the state number that 
was used for that particular sweep, allowing analysis of responses to a particular 
set or combination of stimulus outputs. This is called Dynamic Outputs mode and 

'$GaborFitDemo.s2s|Fit a Gabor function to data in a result view
'===========================================================================================================
'	          CAMBRIDGE ELECTRONIC DESIGN LIMITED, THE SCIENCE PARK, MILTON RD., CAMBRIDGE CB6 0FE, UK
'===========================================================================================================

'Copyright © Cambridge Electronic Design. last modified: Feb 2009.
'Authors:	GPS/GH

'SOFTWARE REQUIRED:
'Spike 2 version 6 or higher
'The script makes use of script functions contained in the file <GHutils.s2s>
'<GHutils.s2s> should be in a folder called <include> inside the Spike2 folder.
'Keep this script <GaborFitDemo.s2s> in the <scripts> folder of the <Spikex> directory (where x stands for the version number).

'HEALTH WARNING:
'The script is a <work in progress> and is offered without guarantees.


'OVERVIEW
'This script is an example of curve-fitting by non-linear regression using the FitNLUser() function.
'The Gabor function used here is a sine wave damped by a Gaussian function.
'The script allows you to generate suitable test data and then apply the curve fitting procedure.
'Alternatively, you could apply the curve fit to your own physiological data.
'An example application for the script would be to detect oscillatory neural activity in auto-correlograms, e.g., Young et al (1992)
'You could also use this script as a template for fitting other functions using non-linear regression by plugging in
'a different user function and modifying the dialogs etc. to match.
'
'REFERENCE
'Young MP, Tanaka K, Yamane S On Oscillating Neuronal Responses in the Visual Cortex of the Monkey  J NeuroPhysiol 67:1464 - 1474



'USER GUIDE
'When you run this script for the first time, a dialog will give you the option to add a short-cut key to the Script Bar.
'If you take this opportunity, you will be able to run the script by clicking on the  button labelled <FitGabor> next time.
'If you don't want a script bar button and you don't want to be "nagged" about it in future, you should comment out the line marked
'with **** below.
'
'The script toolbar has 4 buttons: |Gabor Fit|Browse|Demo File|Quit|
'
'QUIT        simply iconises the log view and closes the script
'
'DEMO FILE   allows you to create a result view containing data generated by mixing the results of a Gabor function with random noise.
'You can select values for the coefficients to use and the amount of noise to add in a dialog.
'The demo data file is generated when you click on OK. Note that the values of the coefficients of the data are also printed below the x-axis.
'
'BROWSE      As an alternative to creating a demo file, you can browse to and open a result view holding data suitable for fitting
'a Gabor function. The previous result view (if any) will be closed before the new one opens.
'
'GABOR FIT   This button will enable after you have created or opened a suitable data file.  When you click on it, a dialog opens where
'you must enter your initial guesses as to the value of the coefficients of the Gabor fit and the x-range to use.
'You can select the X-range by dragging cursors or typing values in the dialog.
'If the initial guesses are not reasonable the curve fitting procedure may not generate a sensible result.
'The script tries to help by choosing plausible values for offset <b> (the mean level in the right-most 25% of the data)
'and amplitude <a>, (the maximum y value in the file). However, the values for the other parameters are entirely up to you.
'The default values are not guaranteed to be sensible.
'
'When you click on OK, the dialog closes and the curve fit is performed. In the case of demo data files, the resulting fit 
'is superimposed on the raw data and a table of results is displayed in the log view. Both the coefficients used to generate the test data
'and the coefficients of the fit are shown.
'
'In the case of "real" data files, i.e., not test data created by this script, the curve is fitted to a copy of the original result view.
'Either way, you are prompted to save data file containing the fitted curve or to re-try with a different initial guess.

'Be aware that fitted curves will be discarded without query if you click on the Close box of result views that have not been
'explicitly saved.

#include <ghutils.s2s>
const demobtn%:=2,newbtn%:=3,fitbtn%:=4;	             'script toolbar button numbers
const on%:=1,off%:=0;
var ditm%;	                                           'dialog item number
var tab%;	                                           'tab size
var rvh%;	                                           'handle of result view
var datach%;	                                        'channel in result view holding data to fit
var fitch%;	                                           'channel to hold the fitted curve
var rootpath$,hotkey$;
var ok%;
var xofs,st,nd;	                                     'xofs and bin number of first bin in result view to fit (i.e. at Cursor(1))
var startndx%;
var Clast[3];	                                        'previous cursor positions

rootpath$:=FilePath$(2);	                            'path to Spike2 application
hotkey$:=Print$("FitGabor|%s\scripts\\GaborFitDemo.s2s|Fit Gabor function to data in result view",rootpath$);
'comment this line to avoid being nagged about adding a button to the Script Bar-----------------------------
HotKeyCheck(hotkey$);
'------------------------------------------------------------------------------------------------------------
View(LogHandle());
tab%:=TabSettings(15);	                               'get current and set new tab size
View(App(3)).WindowVisible(1);	                      'normalise script view before hiding (may reappear if maximised)
HideApps();	                                           'hide unwanted toolbars
DoToolbar();
RestoreApps();	                                        'leave application window as we found it
TabSettings(tab%);	                                  'leave tab setting as we found it
halt

proc DoToolbar();	                                     'set up the script toolbar
ToolbarSet(0,"",Idle%);
ToolbarSet(1,"&Quit",Quit%);
ToolbarSet(demobtn%,"&Demo file||create a data file to test curve fitting",DemoData%);
ToolbarSet(fitbtn%,"&Gabor Fit||Fit a Gabor function to the data",GaborFit%);
ToolbarSet(newbtn%,"&Browse||Open a new result view",NewFile%);
Toolbar("Fit Gabor function demo:",511);
'ToolbarEnable(4,off%);
return;
end;

func Idle%()	                                        'idle disables the Fit button if there is no result view
if ViewKind(rvh%)<> 4 then
	ToolbarEnable(fitbtn%,off%);
else
	ToolbarEnable(fitbtn%,on%);
endif;
return 1;
end;

func Quit%()	                                        'close the script
View(LogHandle()).WindowVisible(2);	                   'iconise the log window
return 0;
end;

func NewFile%()	                                     'open a new result view
if ViewKind(rvh%) = 4 then	                            'result view exists
	View(rvh%);
	FileClose(0);	                                     'close previous file
endif;
rvh%:=FileOpen("",4,2);	                               'open a new one
if rvh% > 0 then
	Window(0,0,50,56);
	WindowVisible(1);
	View(LogHandle());	                               'clear the log
	EditSelectAll();
	EditClear();
else
	'cancel pressed
endif;
return 1;
end;

func DemoData%()	                                     'generate a result view holding data suitable for a Gabor fit
const ndata%:=150;
var ok%;
var binsz:=1.0; 
var title$:="Test Gabor fit";
var xU$:="ms";
var yU$:="Count";
var x[ndata%],y[ndata%],i%, xv;

var lo[5],hi[5],stp[5];
'b              a             g             l           p
lo[0]:=-100.0;  lo[1]:=1.0;    lo[2]:=0.001; lo[3]:=0.01; lo[4]:= -4;	'low and hi limits of range for each parameter in set up dialogs and FitNLUser();
hi[0]:= 100.0;  hi[1]:=1000.0; hi[2]:=0.1;   hi[3]:=1.0;  hi[4]:=4;
stp[0]:= 1; stp[1]:= 1; stp[2]:=0.005 ; stp[3]:=0.01 ; stp[4]:=0.1 ;	'spinner step size used in dialogs

'default values                                       coef# 
var b:=0.002;	                                        '0
var a:=100;	                                           '1
var g:=0.01;	                                        '2
var l:=0.1;	                                           '3
var p:=0.0;	                                           '4
var noiselvl:=5.0;	                                  'default noise level is 5% of (a)
var noiseopt%,noiseofs,noise;
var noiseopt$[2];
noiseopt$[0]:=" rectified"; noiseopt$[1]:=" centred on zero"; 

DlgCreate("Parameters for Test data",70,10);
DlgAllow(511,0,DemoDataCh%);
DlgGroup("Gabor function",1,1,41,1.7);
DlgText("f(x) := b + a * exp( -g * x^2 * 0.5) * Sin(L * x + p))",3,1.8);
DlgReal(1,"Offset           (b)",lo[0],hi[0],0,3.2,stp[0]);
DlgReal(2,"Amplitude     (a)",lo[1],hi[1],0,4.2,stp[1]);
DlgReal(3,"1/sigma^2   (g)", lo[2],hi[2],0,5.2,stp[2]);
DlgReal(4,"Frequency   (L)",lo[3],hi[3],0,6.2,stp[3]);
DlgReal(5,"Phase          (p)",lo[4],hi[4],0,7.2,stp[4]);
DlgReal(6,"Random Noise   (% of a)",0,100.0,0,8.2,1);
DlgList(7,"Noise options:",noiseopt$[],2,0,9.2);
ditm%:=DlgText("Note: Rectified noise will increase offset (b)",4,10.2);
ok%:=DlgShow(b,a,g,l,p,noiselvl,noiseopt%);
if not ok% then return 1; endif;	                      'cancelled

'set up noise level
if noiseopt%=1 then noiseofs:=0.5; else noiseofs:=0; endif; 
noise:=a*noiselvl/100.0;	                            'convert from percent level in dialog

'Generate data
for i% := 0 to ndata%-1 do	                            
   xv := i%;
   x[i%] := i%;
	y[i%] := b + a*exp(-g*xv*xv*0.5)*Sin(l*xv+p)+(Rand()-noiseofs)*noise;
next;

'plot test data in a result view
datach%:=1;
fitch%:=2;
rvh%:=SetResult(2,ndata%,binsz,0,title$,xU$,yU$);	    'create one spare channel for the fitted curve
ok%:=ArrConst(View(rvh%,datach%).[],y[]);	             'paste Gabor function + noise to ch 1 
XTitle$(Print$("b=%.3f; a=%.3f; g=%.3f; l=%.3f; p=%.3f.",b,a,g,l,p));	'add param values to x-axis title 
ChanTitle$(datach%,"Test data");
ChanTitle$(fitch%,"Fit");
ChanHide(fitch%);	                                     'this channel will be used later for the curve fit
Optimise();
Window(0,0,50,56);
WindowVisible(1);

'print the parameters used to create the test data to the log
View(LogHandle());
EditSelectAll();
EditClear();
PrintLog("COEFFICIENTS OF TEST DATA\n");	            
PrintLog("b\t%.f\na\t%.f\ng\t%.f\nl\t%.f\np\t%.f\n",b,a,g,l,p);
PrintLog("\nNOISE\n");
PrintLog("Type:\t%s\n",noiseopt$[noiseopt%]);
PrintLog("Amplitude\t%3.1f%% of (a)\n\n",noiselvl);
return 1;
end;

func DemoDataCh%(item%)	                               'change function of demo data dialog
if DlgValue(7)=0 then
	DlgVisible(1,ditm%);
else
	DlgVisible(0,ditm%);
endif;
return 1;
end;

func GaborFit%()	                                     'fit curve, show statistics and save the results
var i%,ok%,ret%;
const ncoef%:=5;	                                     'number of coefficients in the user function
var guess[5];	                                        'initial guess for b,a,g,l and p
var ndata%;	                                           'array size for the data
var lst%[2],nch%,mn2,done%;
var swap;	                                           'start and end of x- range to fit
var nofitflag%;
var lo[5],hi[5],stp[5];

View(rvh%);
nch%:=ChanList(lst%,1);	                               'count channels in result view
if nch%=2 then mn2:=ArrSum(View(rvh%,2).[]); endif;	 'get mean value of channel 2
if nch%=2 and mn2=0.0 then	                            'if 2 channels with chan 2 empty, assume this is newly created demo data
	datach%:=1; fitch%:=2;
else	                                                 'Real data, i.e. NOT a demo file
	'Select the relevant channel -if there is a choice and make a working copy of the data file
	if nch% > 1 then
		DlgCreate("Select channel",70,60);	             'dialog to choose the channel that holds the data to be fitted
		DlgChan(1,"Channel to fit",1+2048);
		DlgAllow(511);
		ok%:=DlgShow(datach%);
		if not ok% then return 1; endif;	                'cancelled
		fitch%:=nch%+1;
	else
		datach%:=1;	                                     'only one channel in original
		fitch%:=2;
	endif;
	
	'make a copy of the result view with an extra channel for the fit.
	var nbins%,newvh%;
	nbins%:=Len(View(rvh%,1).[]);	                      'number of bins in original
	newvh%:=SetResult(nch%+1,nbins%,BinSize(),BinToX(0),Print$("Copy of %s",WindowTitle$()),XUnits$(),"",XTitle$());	'copy of original result view
	for i%:=1 to nch% do
		ArrConst(View(newvh%,i%).[],View(rvh%,i%).[]);	 'copy data
		ChanTitle$(i%,View(rvh%).ChanTitle$(i%));	       'titles 
		ChanUnits$(i%,View(rvh%).ChanUnits$(i%));	       'units
		DrawMode(i%,View(rvh%).DrawMode(i%));	          'draw mode
		ChanColour(i%,1,View(rvh%).ChanColour(i%,1));	 'channel colours
		ChanColour(i%,2,View(rvh%).ChanColour(i%,2));
	next;
	XRange(View(rvh%).BinToX(XLow()),View(rvh%).BinToX(XLow()));
	Optimise(-1);
	ChanHide(fitch%);
	var xl,yl,xh,yh;
	View(rvh%).WindowGetPos(xl,yl,xh,yh);	             'get position and size of original
	Window(xl,yl,xh,yh);	                               'put copy in same position
	WindowVisible(1);
	View(rvh%);
	FileClose(0);	                                     'close the original
	rvh%:=newvh%;	                                     'use the original view handle for the copy	
endif;

'try to offer reasonable initial guess of offset and amplitude as defaults in the dialog
View(rvh%);
guess[0]:=ChanMeasure(datach%,2,BinToX(MaxTime()*0.75),BinToX(MaxTime()));	'mean level in righthand 25% of trace
guess[1]:=ChanMeasure(datach%,13,BinToX(0),BinToX(MaxTime()));	'Abs Max as initial amplitude
done%:=0;
repeat
	View(LogHandle());	                               'clear the log
	EditSelectAll();
	EditClear();
	View(rvh%);
	if CursorExists(1)=0 then	                         'cursor 1 doesn-t exist
		CursorNew(0,1);
		st:=0;
	else
		Cursor(1,XToBin(st));
	endif;	                                           'add cursors at extremes if not already present
	if CursorExists(2)=0 then
		CursorNew(MaxTime(),2);
		nd:=MaxTime();
	else
		Cursor(2,XToBin(nd));
	endif;
	
	'======================================================================================
	'low and hi limits of range for each parameter in set up dialogs and FitNLUser();	
	'b              a             g             l           p
	lo[0]:=YLow(datach%);  lo[1]:=-2.0*guess[1];    lo[2]:=0.001; lo[3]:=0.01; lo[4]:= -4;	
	hi[0]:= YHigh(datach%);  hi[1]:=2.0*guess[1]; hi[2]:=1000;   hi[3]:=1000.0;  hi[4]:=4;
	stp[0]:= 1; stp[1]:= 1; stp[2]:=0.005 ; stp[3]:=0.01 ; stp[4]:=0.1 ;	'spinner step size used in dialogs	
	'======================================================================================
	FrontView(rvh%);
	DlgCreate(" Initial guesses",70,60);
	DlgAllow(511,GuessDlgIdl%,GuessdlgChnge%);
	DlgGroup("Gabor function",1,1,44,1.7);
	DlgText("f(x) := b + a * exp( -g * x^2 * 0.5) * Sin(L * x + p))",5,1.8);
	DlgReal(1,"Offset           (b)",lo[0],hi[0],0,3.2,stp[0]);
	DlgReal(2,"Amplitude     (a)",lo[1],hi[1],0,4.2,stp[1]);
	DlgReal(3,"1/sigma^2   (g)", lo[2],hi[2],0,5.2,stp[2]);
	DlgReal(4,"Frequency    (L)",lo[3],hi[3],0,6.2,stp[3]);
	DlgReal(5,"Phase          (p)",lo[4],hi[4],0,7.2,stp[4]);
	DlgGroup("X range",1,8.2,44,2);
	DlgText("Start",2,9.2);
	DlgText("End",25,9.2);
	DlgReal(6,12,0,MaxTime(),10,9.2,BinSize());
	DlgReal(7,12,0,MaxTime(),32,9.2,BinSize());
	ok%:=DlgShow(guess[],st,nd);
	if not ok% then return 1; endif;
	if st>nd then
		st:=swap; st:=nd; nd:=swap;	                   'flip start and end;
	endif;
	if XToBin(nd)-XToBin(st)<=Len(guess[]) then
		Message("Not emough data in the x-range.");
		return 1;
	endif;
	xofs:=Trunc(XToBin(st))*BinSize();	                'offset truncated to an exact number of bins
	'x value of the start point of region to fit (global so that the user function knows its value)
	startndx%:=XToBin(st);	                            'bin number of the start opf region to fit (global variable required by user function)
	ndata%:=XToBin(nd)-startndx%;	                      'number of points to fit
	':=Len(View(rvh%,datach%).[]);
	'DO CURVE FITTING
	'Phase 1. Set up the problem. This tells the script language the name of
	'the function to use to get the data values, then sets a range of
	'acceptable values for each argument and a starting guess for each
	'value. It is MOST important that the starting guess is reasonable.
	'If it is not, the fitting may yield stupid results.
	FitNLUser(UserFnc, ncoef%, ndata%);	                'Link user function
	
	'Set initial guess and range for each parameter
	FitNLUser(0, guess[0], lo[0],hi[0]);	             'offset           b
	FitNLUser(1, guess[1], lo[1],hi[1]);	             'amplitude        a
	FitNLUser(2, guess[2], lo[2],hi[2]);	             '1/sigma**2       g
	FitNLUser(3, guess[3], lo[3],hi[3]);	             'frequency        l
	FitNLUser(4, guess[4], lo[4],hi[4]);	             'phase            p
	
	'Phase 2. Solve the problem. This tells the system to improve the initial guess.
	'Each iteration will call the User function <ndata%> times. The result is 0 if it worked.
	'<err> is the sum of squares of the difference between the fitted curve and the raw data
	'<maxI%> maximum number of iterations to do before stopping
	'<iter%> is the number of times that the user function was called
	'<coefs> are the fitted parameters.
	'covar> holds the co-variance matrix
	var coefs[NCOEF%], err, maxI%:=100, iter%, covar[NCOEF%][NCOEF%];
	ret% := FitNLUser(coefs[], err, maxI%, iter%, covar[][]);
	
	'plot the fitted curve over full x -axis (not just the region bracketed with cursors)
	var b,a,g,l,p,xv;
	var sz%;
	sz%:=Len(View(rvh%).[]);
	var x[sz%],y[sz%];
	b:=coefs[0]; a:=coefs[1]; g:=coefs[2]; l:=coefs[3]; p:=coefs[4];
	
	for i% := 0 to sz%-1 do	                            'Generate the curve
		xv :=BinToX(0)+i%*BinSize(datach%);
		x[i%] := BinToX(0)+i%*BinSize(datach%);
		y[i%] := b + a*exp(-g*xv*xv*0.5)*Sin(l*xv+p);
	next;
	ArrConst(View(rvh%,fitch%).[],y[]);	                'paste it to a spare channel in the data file
	DrawMode(fitch%,2);	                               'draw as line
	ChanColour(fitch%,1,16);	                         'red
	ChanShow(-1);
	ChanOrder(fitch%,0,datach%);	                      'overdraw with the original data
	YAxisLock(1,1);	                                  'lock axes
	Optimise(-1);
	
	'Print statistics to the log
	PrintLog("GABOR FIT STATISTICS\n");
	PrintLog("X-Range\t %.4g to %.4g\n",st,nd);
	docase
	case ret%=1 then PrintLog("Fit:\tFit has not converged.\n");
	case ret%=0 then PrintLog("Fit:\tFit has converged.\n");
	case ret%=-1 then PrintLog("Fit:\tFit not improving.\n");
	case ret%<-1 then PrintLog("Fit:\tFit failed.\n");	
	endcase;
	if ret%<-1 then
		nofitflag%:=1;
		DlgCreate("Fit failed",70,60);
		DlgAllow(511);
		DlgText("Retry with different initial guess?",2,1);
		DlgButton(1,"Retry");	                         'rename OK
		ok%:=DlgShow();
		if ok%=0 then
			done%:=1;	                                  'fit failed so, no point in going any further.
		else
			ArrConst(View(rvh%,fitch%).[],0);
			View(rvh%);
			ChanHide(fitch%);
		endif;
	endif;
	if not nofitflag% then
		PrintLog("Error:\t%.3f\t (sum of squares of deviates).\n",err);
		PrintLog("Iterations:\t%d\n",iter%);
		PrintLog("\nCOEFFICIENTS\n");
		PrintLog("\tGUESS\tBEST FIT\n");
		PrintLog("b\t%.3g\t%.3g\na\t%.f\t%.f\ng\t%.f\t%.f\nl\t%.f\t%.f\np\t%.f\t%.f\n",guess[0],b,guess[1],a,guess[2],g,guess[3],l,guess[4],p);
		PrintLog("\nCO-VARIANCE MATRIX\n");
		PrintLog("\t     b\t     a\t     g\t     l\t     p");
		var j%,lbl$[5];
		lbl$[0]:="b"; lbl$[1]:="a"; lbl$[2]:="g"; lbl$[3]:="l"; lbl$[4]:="p"; 
		for i%:= 0 to ncoef%-1 do
			PrintLog("\n %s",lbl$[i%]);
			for j%:=0 to ncoef%-1 do
				PrintLog("\t% 8.3g",covar[i%][j%]);
			next;
		next;
		PrintLog("\n");
		View(LogHandle());
		Window(50,0,100,56);
		FrontView(LogHandle());	                         'make sure that the table is visible
		
		'option to save data and curve fit
		DlgCreate("Query?",70,60);
		DlgAllow(511);
		DlgText("Save fit?",2,1);
		DlgButton(2,"Re-try");
		DlgButton(1,"Save");
		DlgButton(0,"Don't save");
		ok%:=DlgShow();
		docase
		case ok% =1 then
			View(rvh%);
			FileSaveAs("");
			done%:=1;
		case ok%=2 then
			ArrConst(View(rvh%,fitch%).[],0);
			View(rvh%);
			ChanHide(fitch%);
			done%:=0;
		else
			NewsFlash("If you change your mind, you can save the file later via <Save As...> on the File menu.",5,0,60,0);
			done%:=1;
		endcase;
	endif;
until done%=1;
return 1;
end;

func UserFnc(ind%, a[], dyda[])	                      'The user-defined function: f(x) = b + a*exp(-g*x^2*0.5)*Sin(l*x+p);
'ind%:  index into the data
'<a> array holding the current values of the coefficients
'<dyda> array holding the differentials of the function with respect to each coefficient for the current data point
'as calculated by the user function

'dy/db = 1;
'dy/da = exp(-g*x^2*0.5)*Sin(l*x+p)
'dy/dg = -x^2*0.5*a*exp(-g*x^2*0.5)*Sin(l*x+p)
'dy/dl = a*x*exp(-g*x^2*0.5)*Cos(l*x+p)
'dy/dp = a*exp(-g*x^2*0.5)*Cos(l*x+p)
var yv,xi,yi, xsq2, gauss, angle, sine, cosine;
xi := xofs+View(rvh%).BinSize()*ind%;	                'local copy of current x value
yi := View(rvh%,datach%).[startndx%+ind%];	          'local copy of current y value
xsq2 := xi*xi*0.5;
gauss := Exp(-a[2]*xsq2);
angle := a[3]*xi+a[4];
sine := Sin(angle);
cosine := Cos(angle);

dyda[0] := 1;	                                        'dy/db is 1
dyda[1] := gauss*sine;	                               'exp(-g*x^2*0.5)*Sin(l*x+p)
dyda[2] := -xsq2*a[1]*dyda[1];	                      '-x^2*0.5*a*exp(-g*x^2*0.5)*Sin(l*x+p)
dyda[4] := a[1]*gauss*cosine;	                         'a*exp(-g*x^2*0.5)*Cos(l*x+p)
dyda[3] := dyda[4]*xi;	                               'a*x*exp(-g*x^2*0.5)*Cos(l*x+p)
yv := a[0] + a[1]*dyda[1];
'PrintLog("%d,%g,%g,%g\n",ind%,xi,yv,dyda);
return yi-yv;
'returns the difference between current data point and the calculated vale based on the coefficients that were passed in
end;

func HotKeyCheck(btnlabel$)	                         'option to add this script to the Script Bar
'btnlabel$ is the button label for a Script Bar hotkey to give one-click access to the current script
'It must have the format descibed for the <set$> argument in the Spike2 ScriptBar() function.
'This function checks the button labels of current script bar buttons
'If no match is found, then you will get the option to add the button to the Script Bar.
var exists%,done%,btn%,nbtns%;
var get$;

nbtns%:=ScriptBar();	                                  'count script bar buttons
btn%:=0;
exists%:=0;
done%:=0;
repeat
	ScriptBar(btn%,get$);	                            'check labels of existing buttons
	if InStr(get$,btnlabel$)>0 then	                   'button with this name already exists
		exists%:=1;
		done%:=1;
	endif;
	btn%+=1;
until btn%>nbtns% or done%;	                         'until we found a match or all buttons checked
if exists%=0 then
	DlgCreate("Query",70,10);
	DlgText("Add this script to the Script Bar?",2,1);
	DlgButton(0,"No");
	DlgButton(1,"Yes");
	ok%:=DlgShow();
	if ok% then
		AddScriptBarBtn%(btnlabel$);
	endif;
endif;
return;
end;
func GuessdlgChnge%(item%)
DlgToCsr%(item%,6,1,rvh%);
DlgToCsr%(item%,7,2,rvh%);
return 1;
end;

func GuessDlgIdl%()
CsrToDlg(6,1,rvh%);
CsrToDlg(7,2,rvh%);
return 1;
end;

func DlgToCsr%(item%,ditm%,csr%,resvh%)	             'move cursor to match time set in dialog
'This function links a DlgReal() item for an x-coordinate to a cursor position so that the cursor jumps to the new position
'when the dialog item is updated. The function must be called in the Change function of the dialog.

'Arguements:
'item%:	the dialog item that is currently being updated
'ditm%:	the dialog item to link to a cursor position
'csr%:	the number of the cursor to link to the dialog item. The cursor must already exist.
'timevh%:	the handle of the view containing the cursor
var vh%;
vh%:=View();	                                        'remember current view
if ViewKind(resvh%) = 4 then
   View(resvh%);
	if item% =ditm% then	              	
      Cursor(csr%,XToBin(DlgValue(ditm%)));	          'move cursor to new position
	endif;
   Clast[csr%]:=-1;	                                  'force idle routine to update selection
endif;
View(vh%);	                                           'restore previous view
return 1;
end;

proc CsrToDlg(ditm%,csr%,resvh%)	                      'update dialog item when user drags a cursor
'This procedure links a cursor position to a DlgReal() item so that the dialog item updates when the cursor is dragged.
'It must be called in the Idle function of the dialog.

'Arguements:
'ditm%:	the dialog item to link to a cursor position
'csr%:	the number of the cursor to link to the dialog item. The cursor must already exist.
'timevh%:	the handle of the view containing the cursor

var vh%;
vh%:=View();	                                        'remember current view
if ViewKind(resvh%)= 4 then
   View(resvh%);
   if Cursor(csr%) <> Clast[csr%] then	                'if cursor moved since last time we looked...
      DlgValue(ditm%,BinToX(Cursor(csr%)));	          'update dialog to match
      Clast[csr%]:=Cursor(csr%);	                      'remember current position
   endif;
endif;
View(vh%);	                                           'restore previous view
return;
end;

CED

is the most useful and general-purpose way of using multiple states during 
sampling. 

 
Signal sampling configuration 

 Multiple states are enabled by selecting the Multiple frame states checkbox in the 
General tab of the sampling configuration. Once enabled, a States tab is added 
to the configuration from where the mode, number of extra states (state 0 is 
always available as the default sampling state) and ordering during the 
experiment can be defined. Signal can switch between states in numeric or 
random order during an experiment, or run multiple user-defined protocols for 
state sequencing. 

 
States configuration tab 

 The outputs generated by the different states are set-up in the pulses 
configuration dialog, available in the Outputs tab of the sampling configuration. 
States can be selected from the drop-down list in the pulses configuration dialog 
and pulses are added to the frame of outputs by clicking and dragging pulses 
from the palette to the output ‘tracks’ in the graphical display. 



 
Pulses configuration dialog 

 When sampling with multiple states Signal provides a States control toolbar to 
display the current state and allow manual or automated control of the state 
sequencing during the experiment. 

 
States control toolbar 

 The next edition of the eNewsletter will give examples of using multiple states for 
clamping experiments and MagStim control in TMS studies. 

Scripts: Spike2  
Q. I have been using the artefact removal script (ArtRem.s2s) from the website and 

would like to use it on some imported data files. Unfortunately these files have 
been imported as RealWave data and the script expects waveforms. Is there any 
way to run the artefact removal on RealWave channels? 

A. The attached script, ArtRemOverlay.s2s, is an updated version of the 
ArtRem.s2s script, and now removes artefacts from both waveform and 
RealWave channels. The new version creates a duplicate of the original view 
with the sections of data around the selected trigger overdrawn, allowing the user 
to see all of the artefacts together. The main dialog is now interactive with 
cursors in the overdraw view to select the data area for removal. This script 
requires Spike2 version 6 as it makes use of the overdraw display modes. 

Scripts: Signal  
Q. Is it possible to create a new channel in a Signal data file? I would like to create a 

new channel to hold the result of subtracting one channel from another in the 
original frame. 

A. The attached script, ChanArithmetic.sgs, is an example script to perform 
simple arithmetic between two waveform channels in the current frame. It creates 
a new file holding the original waveforms along with the result in a separate 
channel. 

Did you know…? 
Filter banks can be saved and loaded from the FIR and IIR interactive dialogs in Spike2 
and Signal. 

Recent questions   
Q. Is it possible to use active cursors on spiking data? I would like to find the half-

height of action potentials. 

A. Active cursors can be set-up to find features such as peaks, troughs and 
repolarisation % on spike channels drawn as a waveform or WaveMark, in the 
same way as they can on continuous waveform data. For this example, we set 
Cursor 0 to find the peak of the spike, Cursor 1 to find the beginning of the rise 

Back to top 
Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know. 

Back to top 
Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know. 

'$ArtRemOverlay|This script removes stimulus artifacts from waveform or RealWave data channels.  It starts from offset seconds 
'after the trigger times and removes points% number of points, and replaces them with a straight line so it all lines up

'**Further modifications which require at least Spike2 version 6.09**
'This version creates a duplicate of the original view with the section around the trigger events overdrawn
'enabling the user to see the artifacts together. The dialog is now interactive, and there are cursor displayed
'in the overdraw view which can be used to set the area for removal.

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 20/02/09

var data%;                                  'Data file handle
var dataChan%;                              'Waveform / realwave channel for artifact removal
var trigChan%;                              'Trigger channel
var offset:=0;                              'Offset from trigger time
var points%:=4;                             'Number of points to remove
var oOffset,oPoints%;
var ret%;
var dup%;                                   'Duplicate window showing overdrawn 'sweeps'
var dispTime := 0.2;                        'Time range to display in duplicate window
var preTime;                                'Pre trigger time to display in duplicate window
var possTrigChans%[2];                      'Array to check if any available trigger channels in the file
var possWaveChans%[2];                      'Array to check if any suitable waveform/realwave channels available
var oStart,oEnd;                            'Previous cursor positions
var bSize;                                  'Bin size of waveform channel

if App(-1)<609 then Message("This script requires Spike2 version 6.09 or higher"); halt endif;	'Checks for the required version of Spike2

ToolbarVisible(1);
data% := FileOpen("",0,2);
if data% > 0 then
    ChanList(possTrigChans%[],126);         'Check we have possible trigger channel
    if possTrigChans%[0] > 0 then
        trigChan% := possTrigChans%[1];     'Set initial trigger channel to first possible
    else
        Message("No suitable trigger channels in file");
        Halt;
    endif;
    ChanList(possWaveChans%[],513);         'Check we have available waveform/realwave channels in the file
    if possTrigChans%[0] > 0 then
        dataChan% := possWaveChans%[1];     'Set initial trigger channel to first possible
        bSize := BinSize(dataChan%);
    else
        Message("No suitable channels in file");
        Halt;
    endif;
    XRange(0,Maxtime());                    'Show whole file
    Optimise(-1);
    Window(0,0,100,50);
    CursorSet(2);
    CursorLabel(4,1,"Start");
    CursorLabel(4,2,"End");
    FrontView(data%);
    dup% := WindowDuplicate();              'Set duplicate of data window
    Window(0,50,50,100);
    XRange(0,dispTime);
    preTime := dispTime/2.0;                'Set so that trigger point is central
    CursorSet(2);
    CursorLabel(4,1,"[art");
    CursorLabel(4,2,"art]");
    WindowVisible(1);
    ToolbarText("Set time range for artifact rejection with cursors in main view and artifact parameters with cursors and dialog");
    DlgCreate("Artifact removal",50,50);
    DlgChan(1,"Select waveform data channel",513);
    DlgChan(2,"Select trigger channel",2+4+8);
    DlgReal(3,"Enter time (in sec) after trigger to atart removing arftifact", -1000, 1000,28,3,bSize);
    DlgInteger(4,"Enter number of points to remove",1,1000000,28,4,1);
    DlgAllow(1023, DlgIdle%,Change%);                           'Set up dialog idle routine and check function
    ret%:=DlgShow(dataChan%, trigChan%, offset, points%);
    if ret%>0 then
        DelArt%(dataChan%,trigChan%,View(data%).Cursor(1),View(data%).Cursor(2),offset,points%);
    endif;
endif;



Func DlgIdle%()                                                 'Idle routine to run whikle dialog is open
var cOffset,cPoints%;
var wData[1000];                                                'Array used to counts points between cursors
View(data%);
if Cursor(1) <> oStart or Cursor(2) <> oEnd then                'If either cursor has moved
    DoOverDraw%();                                              'Change overdraw display based on new positions
    oStart := View(data%).Cursor(1);                            'Reset reference values
    oEnd := View(data%).Cursor(2);
endif;
View(dup%);
cOffset := Cursor(1)-Cursor(0);                                 'Get current offset from cursor position in overdrawn view
cPoints% := ChanData(dataChan%,wData[],Cursor(1),Cursor(2));    'Get number of points between curors
if cOffset <> oOffset or cPoints% <> oPoints% then              'If changed
    CursorRenumber();                                           'Renumber cursors
    CursorLabel(4,1,"[art");                                    
    CursorLabel(4,2,"art]");
    DlgValue(3,cOffset);                                        'Adjust dialog values to show cursor positions
    DlgValue(4,cPoints%);
    oOffset := cOffset;                                         'Reset reference values
    oPoints% := cPoints%;
endif;
return 1;
end;

Func Change%(item%);                                            'Change function to react to entries in dialog
trigChan% := DlgValue(2);                                       'Get trigger channel
dataChan% := DlgValue(1);                                       'Get waveform / realwave channel
offset := DlgValue(3);                                          'Get offset
points% := DlgValue(4);                                         'Get number of points
bSize := View(data%).BinSize(dataChan%);                        'Get binsize of channel
DlgValue(2003,bSize);                                           'Adjust spinner increment to bin size of channel
View(dup%);
ChanHide(-1);                                                   'Show only required channel
ChanShow(trigChan%);
ChanShow(dataChan%);
DoOverDraw%();                                                  'Update overdrawn view
return 1;
end;


Func DoOverDraw%()                                              'Function to overdraw 'sweeps' within selected range
View(dup%);
dispTime := XHigh()-XLow();                                     'Get time range displayed
preTime := dispTime/2.0;                                        'Set pre trigger time to half of display
ViewTrigger(trigChan%,preTime,0,1,1,0,1);                       'Set triggered display
ViewOverDraw(1,trigChan%,View(data%).Cursor(1),View(data%).Cursor(2));  'Overdraw time range between cursors in main view
CursorActive(1,16,0,Print$("Cursor(0)+%f",offset));                     'Set cursor 1 to move to offset position
CursorActive(2,14,dataChan%,"Cursor(1)","MaxTime()","",points%);        'Set cursor 2 to move n points after cursor 1
CursorSearch(1);                                                        'Move cursors to positions
oOffset := offset;                                                      'Reset reference values 
oPoints% := points%;
CursorVisible(0,0);
return 1;
end;


func DelArt%(dataCh%, trigCh%, locSTime, locETime, offset, points%)
var mem%;
var time;
var arr[points%];
var bin;
var list%[33];
var title$;
var chType%;
bin:=Binsize(dataCh%);                                          
title$:=ChanTitle$(dataCh%);                                    
chType% := ChanKind(dataCh%);                                           'Get channel type (waveform or realwave)
if chType% = 1 then                                                     'Set memory channel of same type
    mem%:=MemChan(1,0,bin);
else
    mem% := MemChan(9,0,bin);
endif;
ChanTitle$(mem%, title$);                                               'Set title, scale and offset to match source channel
ChanScale(mem%,ChanScale(dataCh%));
ChanOffset(mem%,ChanOffset(dataCh%));
MemImport(mem%,dataCh%,0,MaxTime());                                    'Import data from source channel
time := locSTime-BinSize();
repeat
    time:=NextTime(trigCh%, time);                                      'Step through each trigger event
    if time>0 and time<locETime then                                    
        MakeSlope%(arr[], ChanValue(mem%, time+offset), ChanValue(mem%, time+offset+points%*bin));  'Remove artifact
        MemSetItem(mem%, 0, time+offset, arr[]);
    endif;
    until time<0 or time>locETime;
ChanList(list%[], 128);                                                 'Find next available empty channel
if list%[0]>0 then
    ChanTitle$(dataCh%, Left$(title$,6)+"(B)");                         'Save modified channel
    MemSave(mem%, list%[1]);
    ChanDelete(mem%);
    Optimise(list%[1]);
    ChanShow(list%[1]);
    View(data%);
    Optimise(list%[1]);
    ChanShow(list%[1]);
endif;
end;

func MakeSlope%(arr[], sVal, eVal);
var len%;
var inc;

len%:=Len(arr[]);
inc:=(eVal-sVal)/len%;

ArrConst(arr[], inc);
arr[0]:=0;
ArrIntgl(arr[]);
ArrAdd(arr[], sVal);
return 1;
end;

'$ChanArithmetic|Creates a duplicate view containing a result channel of simple arithmetic between 2
'selected waveform channels

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 02/03/09

var file%;					'Handle for source file 
var dupFile%;				'Handle for result file 
var math%;                  'Math
var binSz,offset,noPts;     'Settings from source file
var chan1%,chan2%;            'Selected channels
var yunits1$,yunits2$,yTitle1$,yTitle2$;
var xUs$;

file%:=FileOpen("",0,0);	'Open file	
if file%<0 then				'Check file opened
	Message("Failed to open file");
	Halt;
endif;
View(file%);
Window(0,0,50,100);			'Position window
Interact("Select frame for channel arithmetic and click OK",1023);
FrontView(file%);			'Bring to front
Optimise(-1);				'Optimse the data

DlgCreate("Channel Arithmetic");
DlgList(1,"Function","Add|Subtract|Divide|Multiply");
DlgChan(2,"Channel",1);
DlgChan(3,"Channel",1);
DlgShow(math%,chan1%,chan2%);

binSz:=BinSize(chan1%);
offset:= MinTime();
noPts:=(1/binSz)*(MaxTime()-offset);
yunits1$:= ChanUnits$(chan1%);
yTitle1$:= ChanTitle$(chan1%);
yunits2$:= ChanUnits$(chan2%);
yTitle2$:= ChanTitle$(chan2%);
xUs$:= XUnits$();

MakeResFile();

Proc MakeResFile();
View(file%);

dupFile%:=SetMemory(3,noPts,binSz,offset,0,0,0,"Arithmetic View",xUs$);
ChanUnits$(1,yUnits1$);
ChanTitle$(1,yTitle1$);
ChanUnits$(2,yUnits2$);
ChanTitle$(2,yTitle2$);

ArrConst(View(dupFile%,1).[],View(file%,chan1%).[]);
ArrConst(View(dupFile%,2).[],View(file%,chan2%).[]);
ArrConst(View(dupFile%,3).[],View(file%,chan2%).[]);

docase
    case math% = 0 then
        ArrAdd(View(dupFile%,3).[],View(dupFile%,1).[]);
        ChanUnits$(3,yUnits1$);
        ChanTitle$(3,"Result Add");
    case math% = 1 then
        ArrSub(View(dupFile%,3).[],View(dupFile%,1).[]);
        ChanUnits$(3,yUnits1$);
        ChanTitle$(3,"Result Sub");
    case math% = 2 then
        ArrMul(View(dupFile%,3).[],View(dupFile%,1).[]);
        ChanUnits$(3,yUnits1$);
        ChanTitle$(3,"Result Mul");
    case math% = 3 then
        ArrDiv(View(dupFile%,3).[],View(dupFile%,1).[]);
        ChanUnits$(3,yUnits1$);
        ChanTitle$(3,"Result Div");
endcase;
    
Window(50,0,100,100);									'Position window
Optimise(-1);
WindowVisible(1);
end;

and Cursor 2 to find the 50% re-polarisation between the two when the Cursor 0 
position is used as the 100% value.  

 
Attached is a resource file, Half-height example.s2r, holding the active 
cursor settings described here which can be applied to the example 
Extracellular spikes.smr data file in the Spike6\data directory using the 
Resource files option from the File menu in Spike2.  

User group      
Try the new CED Forums bulletin board for software and hardware support 

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 

